NASA TECHNICAL MEMORANDUM



NASA TM X-3089

NASA TM X-3089

PROTON TISSUE DOSE FOR THE BLOOD FORMING ORGAN IN HUMAN GEOMETRY: ISOTROPIC RADIATION

by G. S. Khandelwal and John W. Wilson

Langley Research Center

Hampton, Va. 23665

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • DECEMBER 1974

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TM X-3089		

4. Title and Subtitle	5. Report Date
PROTON TISSUE DOSE FOR THE BLOOD FORMING ORGAN IN HUMAN GEOMETRY: ISOTROPIC RADIATION	December 1974
	6. Performing Organization Code

7. Author(s)	8. Performing Organization Report No.
G. S. Khandelwal and John W. Wilson	L-9628
	10. Work Unit No.
9. Performing Organization Name and Address	502-21-32-01
NASA Langley Research Center Hampton, Va. 23665	11. Contract or Grant No.
	13. Type of Report and Period Covered
12. Sponsoring Agency Name and Address	Technical Memorandum
National Aeronautics and Space Administration Washington, D.C. 20546	14. Sponsoring Agency Code

15. Supplementary Notes

G. S. Khandelwal is a Professor of Physics, Old Dominion University, Norfolk, Va.

16. Abstract

A computer program is described which calculates doses averaged within five major segments of the blood forming organ (BFO) in the human body taking into account self-shielding of the detailed body geometry and nuclear star effects for proton radiation of arbitrary energy spectrum (energy less than 1 GeV) and isotropic angular distribution. The dose calculation includes the first term of an asymptotic series expansion of transport theory which is known to converge rapidly for most points in the human body. The result is always a conservative estimate of dose and is given as physical dose (rad) and dose equivalent (rem).

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Radiation Doses Nuclear reaction	Unclassified – Unlimited STAR Category 29

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	19	$3.25

*For sale by the National Technical Information Service, Springfield, Virginia 22151

PROTON TISSUE DOSE FOR THE BLOOD FORMING ORGAN IN HUMAN GEOMETRY: ISOTROPIC RADIATION

By G. S. Khandelwal* and John W. Wilson
Langley Research Center

SUMMARY

A computer program is described which calculates doses averaged within five major segments of the blood forming organ (BFO) in the human body taking into account self-shielding of the detailed body geometry and nuclear star effects for proton radiation of arbitrary energy spectrum (energy less than 1 GeV) and isotropic angular distribution. The dose calculation includes the first term of an asymptotic series expansion of transport theory which is known to converge rapidly for most points in the human body. The result is always a conservative estimate of dose and is given as physical dose (rad) and dose equivalent (rem).

INTRODUCTION

A necessary requirement for radiation shield and dosimeter design and for space mission analysis is a reliable method of calculating the anticipated dose distribution in the human body for the pertinent radiation environment. Customarily, the human body is approximated by a simple geometric object (such as a sphere, slab, or cylinder) with resultant disagreement among the various approximations (ref. 1) and consequent disagreement on shield and dosimeter design and impact on mission objectives.

Described herein is a computer program which calculates proton dose averaged over five major segments (upper limbs, lower limbs, upper trunk, lower trunk, and skull) of the blood forming organ (BFO) by treating the human body geometry in detail (refs. 2 and 3) but assuming isotropicity of the incident primary particles. The calculation uses an approximate form of transport theory in which nuclear star effects are incorporated (refs. 4 and 5). The output of the program is in terms of physical dose (rad) and dose equivalent (rem).

---

*Professor of Physics, Old Dominion University, Norfolk, Va.

SYMBOLS

$D(\vec{X})$	dose at the point $\vec{X}$, rad or rem
E	incident proton energy, MeV
$f_X(t)$	areal density distribution function, $\frac{1}{g/cm^2}$
$R(t,E)$	dose response at depth t for protons of energy E, $\frac{\text{rad or rem}}{\text{proton/cm}^2}$
t	depth from surface or areal density, g/cm^2
$\vec{X}$	vector locating point within the body, g/cm^2
$\vec{\Omega}$	vector denoting direction of motion of radiation, dimensionless
$\phi(\vec{\Omega},E)$	proton differential fluence spectrum, protons/cm^2-MeV-sr

THEORY

An asymptotic expansion for the dose at a point $\vec{X}$ is derived in reference 5. The first term of the expansion is an accurate approximation of the dose given by

$$D(\vec{X}) = \int_0^\infty \int_\Omega R\left[t(\vec{\Omega}),E\right]\phi(\vec{\Omega},E)\, d\vec{\Omega}\, dE \tag{1}$$

where the errors are second order in the expansion parameter δ, which is the beam width. In equation (1), $\phi(\vec{\Omega},E)$ is the proton fluence spectrum in the direction $\vec{\Omega}$ with energy E, $R(t,E)$ is the fluence-to-dose conversion factor at the depth t for normal incident protons on a slab, and $t(\vec{\Omega})$ is the distance from the body surface to the point $\vec{X}$ along the direction $\vec{\Omega}$. When the proton fluence spectrum $\phi(\vec{\Omega},E)$ is isotropic, equation (1) can be written as

$$D(\vec{X}) = \int_0^\infty \phi(E) \int_\Omega R\left[t(\vec{\Omega}),E\right] d\vec{\Omega}\, dE \tag{2}$$

since ϕ is no longer dependent on $\vec{\Omega}$ $\left(\text{that is, } \phi(\vec{\Omega},E) \equiv \phi(E)\right)$.

It is advantageous to replace the integral over $\vec{\Omega}$ by an integral over the depth t of tissue. This can be accomplished by the correspondence between the fractional solid angle and the depth t of tissue. Thus, the above integral can be written as

$$D(\vec{X}) = 4\pi \int_0^\infty \phi(E) \int_0^\infty R(t,E)\, f_X(t)\, dt\, dE \tag{3}$$

with

$$\int_0^\infty f_X(t)\, dt = 1 \tag{4}$$

where $f_X(t)$ is the areal density distribution function for the point $\vec{X}$. The quantity $f_X(t)\, dt$ is the fraction of the solid angle for which the distance to the surface from the point $\vec{X}$ lies between t and $t + dt$. The areal density distribution functions of Billings et al. (table 1) are discussed in reference 3 (see also refs. therein and in particular their ref. 23) and differ markedly from their original data in reference 2. The function $R(t,E)$ is given in references 4 and 5. In the program, the integrals over areal density and energy are performed using the Gauss-Legendre quadrature formula (ref. 6).

PROGRAM DETAILS

The data for the program are self-contained except that the user supplies a subroutine to generate the primary fluence spectrum (protons/cm^2-MeV-sr). A sample spectrum is supplied with the program for initial checkout. The output for this sample calculation is given in tables 1 and 2.

In the program, an intermediate function $R_X(E)$ is calculated by evaluating

$$R_X(E) = \int_0^\infty R(t,E)\, f_X(t)\, dt \tag{5}$$

after which the dose is evaluated as

$$D(\vec{X}) = 4\pi \int_0^{E_{max}} R_X(E)\, \phi(E)\, dE \tag{6}$$

where E_{max} is restricted to 1 GeV. A brief explanation of the principal subroutines is given below.

There are three basic functions to be generated for the calculation. These are the areal density distribution function, the fluence-to-dose conversion factors for normal incidence on a slab, and the incident fluence spectrum. Two numerical integrations are then performed to evaluate first equation (5) and then equation (6). These calculations are performed in the following subroutines:

Subroutine	Purpose
APDF	Generates areal density distribution for five segments of BFO and BFO average
ARES	Generates the fluence-to-dose conversion factors R(t,E)
SPECTØR	Generates the incident fluence spectrum (protons/cm^2-MeV-sr)
RØFE	Generates the function $R_X(E)$ given by equation (5). The entry point RTAB generates a table of $R_X(E)$ for interpolation.
BDØS	Generates the dose $D(\vec{X})$ at point $\vec{X}$ by computing the final integral over energy as indicated by equation (6).

The program requires no input other than the incidence spectrum supplied by the user.

A listing of the program is found in the appendix. The language is FORTRAN IV and written for the CDC 6000 series computers. Although it was originally programed on a 60-bit machine, round-off error on shorter word machines should not be a problem for these calculations. The incident spectrum in the list is

$$\phi(E) = \frac{1}{E^5}$$

with a sample output from the program given in tables 1 and 2. This sample calculation requires 112 seconds of CPU time on a CDC 6600 and 25 000 (octal) locations.

Langley Research Center,
National Aeronautics and Space Administration,
Hampton, Va., October 23, 1974.

APPENDIX

PROGRAM LISTING OF THE DOSE CALCULATION

```
      PROGRAM GSK(INPUT,OUTPUT,TAPE3=OUTPUT)

      COMMON/ALPHA/ALP
      REAL DOSE(6,2)
      ALP=1.
      ALP=5.
      CALL INITIAL
      CALL BDOS(DOSE)
      PRINT 1
      PRINT 2 ,(DOSE(I),I=1,6)
      PRINT 3,(DOSE(I),I=7,12)
    1 FORMAT(30X,*              ARMS        LEGS   HI TRUNK   LO TRUNK      SKU
     1LL    AVE BFO*//)
    2 FORMAT(30X,*    REM*,6E10.1//)
    3 FORMAT(30X,*    RAD*,6E10.1////)
      END


      SUBROUTINE INITIAL

      REAL AD(25),PDF(6)
      DATA AD /1.1,1.325,1.595,1.92,2.3,2.75,3.3,4.,4.8,5.75,6.95,8.35,
     110.05,12.,14.5,17.5,21.,25.5,30.5,36.5,44.,53.,63.5,76.,91.5/
      A=1.
      CALL INTAPDF
      PRINT 4
    4 FORMAT(1H1,/////)
      PRINT 5
    5 FORMAT(*     THE BILLINGS AND LANGLEY AREAL DENSITY DISTRIBUTION FU
     1NCTIONS ARE*,////)
      PRINT 3
    3 FORMAT(*       THICKNESS              ARMS              LEGS         HI TRU
     1NK          LO TRUNK            SKULL          AVE BFO*,//)
      DO 1 I=1,25
      CALL APDF(AD(I),PDF)
    1 PRINT 2,AD(I),PDF
    2 FORMAT(7E15.3)
      PRINT 4
      CALL RTAB(A)
      RETURN
      END
```

APPENDIX – Continued

```
      SUBROUTINE BDOS(DOSE)

C     THIS SUBROUTINE COMPUTES THE DOSE AT POINT X BY COMPUTING THE
C     INTEGRAL OVER ENERGY ,THE INTEGRAND BEING THE PRODUCT OF
C     THE INCIDENT FLUENCE SPECTRUM AND THE DOSE RESPONCE
C     AT POINT X, RSUBX OF E
      REAL ANS(6,2),A(6,2),DOSE(6,2)
      EXTERNAL POX
      THK=1.101
      CALL AMI(THK,EMIN)
      EMIN=EMIN+.5
      EOTH=EMIN+10.
      M1=M2=23
      CALL MGAUSS(EMIN,EOTH,M1,DOSE,POX,A,12)
      CALL MGAUSS(EOTH,1000.,M2,ANS,POX,A,12)
      DO 1 L=1,6
      DO 1 I=1,2
    1 DOSE(L,I)=DOSE(L,I)+ANS(L,I)
      RETURN
      END


      SUBROUTINE POX(X,F)

C     THIS SUBROUTINE COMPUTES THE INTEGRAND NEEDED IN
C     COMPUTING DOSE,' D(X) AT POINT X
      REAL F(12),R(12)
      E=X
      CALL SPECTOR(SPECTRA,E)
      SPECTRA=SPECTRA*12.566
      CALL ROFE(R,E)
      DO 1 I=1,12
    1 F(I)=R(I)*SPECTRA
      RETURN
      END


      SUBROUTINE SPECTOR(SPECTRA,E)

C     THIS SUBROUTINE GENERATES THE INCIDENT FLUENCE
C     SPECTRUM (P/CM2-SR-MEV)
      COMMON/ALPHA/ALP
      SPECTRA=1./E**ALP
      RETURN
      END
```

APPENDIX – Continued

```
      SUBROUTINE ROFE(R,T)

C     THIS SUBROUTINE GENERATES RAD AND REM DOSE RESPONSES AT POINT
C     X, R SUB X OF E, FOR FIVE SEGMENTS OF EFO AND EFO AVERAGE
C     THE ENTRY POINT RTAB GENERATES A TABLE OF DOSE RESPONSES AT
C     POINT X, R SUB X OF E, FOR INTERPOLATION
      COMMON /ENERGY/ XENER
      REAL R(12),RT(50,12),C(4,50,12),TAB(50)
      REAL ANS(12),A(12)
      DATA TAB/34.,40.,50.,60.,70.,80.,90.,100.,110.,120.,130.,150.,170.
     1,200.,250.,300.,400.,500.,600.,700.,800.,900.,1000./
      EXTERNAL FOX
      DO 101 I=1,12
  101 CALL SPLIN(TAB,RT(1,I),NPT,C(1,1,I),T,R(I))
      RETURN
      ENTRY RTAB
      NPT=23
      DO 1 I=1,NPT
      XENER=TAB(I)
      ENER=XENER
      UENER=1010.
      CALL RANGE(R,ENER,UENER,A,B)
      TMIN=1.101
      FOTH=R
      COTH=R
      TOTH=R
      IF(FOTH.GE.43.837)FOTH=43.837
      IF(TOTH.GE.5.)TOTH=5.
      M1=20
      CALL MGAUSS(TMIN,TOTH,M1,ANS,FOX,A,12)
      DO 11 J=1,12
   11 RT(I,J)=ANS(J)
      IF(TOTH-5.)1,3,1
    3 IF(COTH.GE.20.)COTH=20.
      CALL MGAUSS(TOTH,COTH,M1,ANS,FOX,A,12)
      DO 12 J=1,12
   12 RT(I,J)=RT(I,J)+ANS(J)
      IF(COTH-20.)1,4,1
    4 CALL MGAUSS(COTH,FOTH,M1,ANS,FOX,A,12)
      DO 13 J=1,12
   13 RT(I,J)=RT(I,J)+ANS(J)
    1 CONTINUE
      DO 10 J=1,12
   10 CALL SPLCN(TAB,RT(1,J),NPT,C(1,1,J))
      RETURN
      END
```

APPENDIX – Continued

```
      SUBROUTINE FOX(X,F)

C     THIS SUBROUTINE COMPUTES THE INTEGRAND FOR RAD AND REM DOSE
C     RESPONSES AT POINT X, R SUB X OF E
      REAL F(6),PDF(6)
      COMMON /ENERGY/ XENER
      T=X
      ENER=XENER
      CALL ARES(T,ENER,REM,RAD)
      CALL APDF(T,PDF)
      DO 1 I=1,6
      F(I+6)=RAD*PDF(I)
    1 F(I)=REM*PDF(I)
      RETURN
      END


      SUBROUTINE  APDF(AD,PDF)

C  THIS PROGRAM GENERATES THE BILLINGS AND LANGLEY AREAL DENSITY
C  DISTRIBUTION FUNCTION PDF(I) FOR FIVE SEGMENTS OF BFO (I=1,5) AND
C  THE AVERAGE OVER BFO (I=6)
      REAL ADT(25),PDFT(6,25),TPDF(25,6),C(4,25,6),PDF(6),FRAC(6)
      DATA ADT/1.1,1.325,1.595,1.92,2.3,2.75,3.3,4.,4.8,5.75,6.95,8.35,
     110.05,12.,14.5,17.5,21.,25.5,30.5,36.5,44.,53.,63.5,76.,91.5/
      DATA TPDF/ .0976, .7284, .5607,1.3730, .8292, .9496, .5745, .5247,
     1  .7562,2.1116,1.3975, .8003, .4134, .1572, .1679, .1500, .2320,
     2  .1480, .1197, .0567, .0445, .0174, .0124, .0071,0.0000,0.0000,
     30.0000,0.0000, .8932, .8422, .9187,1.6069,2.1389,2.0703,2.6675,
     42.6216,2.0224,1.7249,1.4707, .9370, .8303, .4695, .2561, .1667,
     6  .1494, .0607, .0374, .0248, .0218,0.0000, .0937, .2437, .5277,
     6  .6544, .9013, .5971,1.1801,1.1543,1.3451,1.2319,1.4042,1.2816,
     71.5885, .9718, .9593, .6294, .5317, .3994, .2244, .1144, .0638,
     8  .0359, .0291, .0128,0.0000,0.0000,0.0000,0.0000,0.0000,0.0000,
     90.0000, .0313, .0703, .2813, .7045, .9231, .9458, .8355, .9625,
     11.0896, .7021, .5984, .2775, .1600, .1295, .0703, .0375, .0131,
     1  .0036,0.0000,0.0000,0.0000, .0137, .3431, .7471, .7570, .5777,
     2  .3586, .4682, .4310, .5302, .7225, .5201, .3307, .3214, .2258,
     3  .0588, .0175, .0143, .0137, .0139, .0080, .0043, .0057,0.0000,
     4  .1913, .9722,1.1022,3.2638,3.3198,3.2225,3.9725,4.2468,4.9210,
     57.1465,6.7867,5.7726,5.0814,3.8929,3.4752,2.5375,1.8904,1.0986,
     6  .6850, .4636, .2532, .1362, .0836, .0609,0.0000/
      DATA FRAC/ .1374, .3296, .2606, .1920, .0816,1.0014/
      DO 1 I=1,6
      CALL SPLIN(ADT,TPDF(1,I),25,C(1,1,I),AD,PDF(I))
      IF(PDF(I).LT.0.) PDF(I)=0.
    1 CONTINUE
      RETURN
      ENTRY INTAPDF
      DO 3 I=1,6
      DO 3 J=1,25
    3 TPDF(J,I)=TPDF(J,I)/(100.*FRAC(I))
      DO 2 I=1,6
    2 CALL SPLCN(ADT,TPDF(1,I),25,C(1,1,I))
      RETURN
      END
```

APPENDIX – Continued

```
      SUBROUTINE RANGE(R,ENER,UENER,A,B)

C     THIS SUBROUTINE COMPUTES PROTON RANGE IN TISSUE
    9 IF(ENER-10.)1,1,3
    3 IF(ENER-30.)6,6,11
   11 IF(ENER-100.)13,13,12
   12 IF(ENER-300.)22,22,23
   23 IF(ENER-500.)17,17,25
   25 IF(ENER-UENER)69,69,19
    1 A=.002245
      B=1.698
      GO TO 60
    6 A=.001956
      B=1.8
      GO TO 60
   13 A=.0019775
      B=1.8
      GO TO 60
   22 A=.00276624
      B=1.729
      GO TO 60
   17 A=.0054847
      B=1.61
      GO TO 60
   69 A=.0125462
      B=1.477
      GO TO 60
60    CONTINUE
      R=A*ENER**B
      RETURN
 19     WRITE(3,26) ENER
26    FORMAT(1HC,1E15.6)
      END
```

```
      SUBROUTINE AMI(EX,ENER)

C     THIS SUBROUTINE COMPUTES THE CUTOFF ENERGY FOR A GIVEN THICKNESS
C     OF TISSUE
    9 IF(EX-.121)1,1,3
    3 IF(EX-.89163)6,6,11
   11 IF(EX-7.8726)13,13,12
   12 IF(EX-53.068)22,22,23
   23 IF(EX-121.48)17,17,25
   25 IF(EX-340.)69,69,5
    1 A=.002245
      B=1.698
      GO TO 60
    6 A=.001956
      B=1.8
      GO TO 60
   13 A=.0019775
      B=1.8
      GO TO 60
   22 A=.00276624
      B=1.729
      GO TO 60
   17 A=.0054847
      B=1.61
      GO TO 60
   69 A=.0125462
      B=1.477
      GO TO 60
   60 ENER=(EX/A)**(1./B)
      RETURN
    5 CALL EXIT
      END
      SUBROUTINE AEDUC(A,B,EX,ENERP,ENER)
C     THIS SUBROUTINE COMPUTES THE RESIDUAL ENERGY OF PROTON
C     AFTER IT TRAVERSES TISSUE OF A GIVEN THICKNESS
      AR=B
      AA=EX/A
      AAR=AA**(1./AR)
   25 IF(ENER-AAR)26,26,27
   26 ENERP=0.
      RETURN
   27 ENERP=(ENER**(AR)-AA)**(1./AR)
      RETURN
      END
```

```
      SUBROUTINE ANTER(ENER,POQ,PTWQ,PONQ,PO,PTW,PON,R)

C     THIS SUBROUTINE GENERATES THE VALUES OF THE PARAMETERS
C     OF THE ANALYTIC FITS OF THE MONTE CARLO RESULTS
      DIMENSION E(20),AQ(20),A2(20),AA0(20),AA2(20),C(4,20)
      DATA E/30.,40.,50.,60.,70.,80.,100.,120.,150.,170.,200.,
     1250.,300.,400.,500.,600.,700.,800.,900.,1000./
      DATA AQ/1.47,1.475,1.480,1.485,1.49,1.495,1.50,1.505,1.525,1.537,1
     1.554,1.6,1.68,1.77,2.05,2.43,3.21,3.70,4.60,5.75/
      DATA A2/0.0,.001,.002,.003,.0045,.006,.01,.019,.04,.058,.0729,.057
     1,.0408,.0305,.026,.0225,.0177,.016,.014,.012/
      DATA AA0/1.042,1.055,1.069,1.080,1.090,1.1,1.12,1.148,1.159,1.172,
     11.190,1.205,1.22,1.24,1.278,1.322,1.4,1.458,1.56,1.67/
      DATA AA2/.19,.175,.16,.149,.143,.132,.0996,.07,.0568,.0495,.0438,.
     10375,.0314,.0228,.0188,.0156,.01385,.013,.0125,.0118/
      CALL SPLCN(E,AQ,20,C)
      CALL SPLIN(E,AQ,20,C,ENER ,POQ)
      CALL SPLCN(E,A2,20,C)
      CALL SPLIN(E,A2,20,C,ENER ,PTWQ)
      CALL SPLCN(E,AA0,20,C)
      CALL SPLIN(E,AA0,20,C,ENER ,PO)
      CALL SPLCN(E,AA2,20,C)
      CALL SPLIN(E,AA2,20,C,ENER ,PTW)
      FUNC=PTWQ*R
      PONQ=(EXP(FUNC)-POQ)/R
      FUNC=PTW*R
      PON=(EXP(FUNC)-PO)/R
      RETURN
      END


      SUBROUTINE AF(STOPP,QALF,AQ,BQ)

C     THIS SUBROUTINE COMPUTES THE QUALITY FACTOR AS A FUNCTION OF
C     LINEAR ENERGY TRANSFER
      IF(STOPP-35.)11,11,12
   11 QALF=1.
      RETURN
   12 QALF=AQ*(STOPP**(BQ))
      RETURN
      END


      SUBROUTINE ATOPP(ENER,STOPP)

C     THIS SUBROUTINE COMPUTES THE STOPPING POWER FOR PROTON IN TISSUE
    2 ZETA=ENER/938.211
      BETAS=((ZETA*(ZETA+2.))/((ZETA+1.)**2))
      WBE=((1.0222U1*10.**(6))*BETAS)/(1.-BETAS)
      FBET=ALOG(WBE)-BETAS
      STOPP=.30726148*(-2.2378342+.529726*FBET)/BETAS
      RETURN
      END
```

APPENDIX – Continued

```
      SUBROUTINE AROB(PROB,ENERP,UENER,ENER,DENER)

C     THIS SUBROUTINE COMPUTES THE PROBABILITY OF A NUCLEAR REACTION NOT
C     OCCURING BEFORE A GIVEN THICKNESS OF TISSUE IS TRAVERSED
      ENER=ENERP
      IPRO=1
    9 IF(ENER-1.6)63,64,64
   64 IF(ENER-9.5)66,66,65
   65 IF(ENER-55.)67,67,68
   68 IF(ENER-140.)69,69,70
   70 IF(ENER-490.)71,71,72
   72 IF(ENER-UENER)73,73,19
   63 AO=1.
      AON=0.
      ATW=0.
      ATH=0.
      AFO=0.
      GO TO 4
   66 AO=.95500944
      AON=.54524166
      ATW=-2.3204599
      ATH=4.1216447
      AFO=-2.5963624
      GO TO 4
   67 AO=.88724554
      AON=.9798068
      ATW=13.502561
      ATH=-175.26228
      AFO=418.86144
      GO TO 4
   69 AO=.2980451
      AON=133.932
      ATW=-.12453671*10.**(5)
      ATH=.55578849*10.**(6)
      AFO=-.96458432*10.**(7)
      GO TO 4
   71 AO=-.67860829
      AON=.77055714*10.**(3)
      ATW=-.17505698*10.**(6)
      ATH=.19611718*10.**(8)
      AFO=-.86413965*10.**(9)
      GO TO 4
   73 AO=.4793539
```

```
      AON=-.14555836*10.**(4)
      ATW=.14942072*10.**(7)
      ATH=-.55637619*10.**(9)
      AFO=.75787641*10.**(11)
      GO TO 4
    4 CFE=1./ENER
      CSE=CFE*CFE
      CTE=CSE*CFE
      CFO=CSE*CSE
      YF=AO+AON*CFE+ATW*CSE+ATH*CTE+AFO*CFO
      IF(IPRO-1)5,7,5
    7 IPRO=2
      YNU=YF
      ENER=DENER
      GO TO 9
 19     WRITE(3,26) ENER
26    FORMAT(1H0,1E15.6)
      CALL EXIT
    5 PROB=YF/YNU
      RETURN
      END


      SUBROUTINE ARES(XINT,ENER,RESF,WRESF)

C     THIS PROGRAM GENERATES THE FLUENCE TO DOSE CONVERSION FACTORS
C     R(T,E) AS A FUNCTION OF TISSUE   THICKNESS AND ENERGY
      EX=XINT
      AQ=.06
      BQ=.8
      DENER=ENER
      UENER=1010.
56    CONTINUE
      CALL RANGE(R,ENER,UENER,A,B)
      CALL ANTER(ENER,POQ,PTWQ,PONQ,PO,PTW,PON,R)
      CALL AEDUC(A,B,EX,ENERP,ENER)
      IF(ENERP)34,33,34
33     CONTINUE
      RESF=0.
      WRESF=0.
      ENER=DENER
      RETURN
34    CONTINUE
      CALL AROB(PROB,ENERP,UENER,ENER,DENER)
      ENER=ENERP
      CALL ATOPP(ENER,STOPP)
      IF(STOPP)33,33,1
    1 IF(STOPP-2700.)2,2,33
    2 CALL AF(STOPP,QALF,AQ,BQ)
   22 PES=PROB*STOPP*QALF
      COREQ=(POQ+PONQ*XINT)/(EXP(PTWQ*XINT))
      COREC=(PO+PON*XINT)/(EXP(PTW*XINT))
      RESF=PES*COREQ*1.6E-8
      WRESF=STOPP*PROB*COREC*1.6E-8
      ENER=DENER
      RETURN
      END
```

APPENDIX – Continued

```
      SUBROUTINE SPLIN(X,Y,M,C,XINT,YINT)

C     THIS SUBROUTINE INTERPULATES AMONG THE DATA POINTS TO PROVIDE VALVES
C     ON SUBINTERVALS AFTER CONSTANTS ARE CALCULATED BY SUBROUTINE SPLCN
      DIMENSION X(50),Y(50),C(4,50)
      DATA IP/0/
      IF(XINT-X(1))7,1,2
    1 YINT=Y(1)
      RETURN
    2 K=1
    3 IF(XINT-X(K+1))6,4,5
    4 YINT=Y(K+1)
      RETURN
    5 K=K+1
      IF(M-K)7,7,3
    6 YINT=(X(K+1)-XINT)*(C(1,K)*(X(K+1)-XINT)**2+C(3,K))
      YINT=YINT+(XINT-X(K))*(C(2,K)*(XINT-X(K))**2+C(4,K))
      RETURN
    7 PRINT 101
  101 FORMAT(*   OUT OF RANGE FOR INTERPOLATION  *)
      IP=IP+1
      IF(IP.GT.50) GO TO 10
      RETURN
   10 CONTINUE
      A=ALOG(-1.)
      END
```

APPENDIX – Continued

```
      SUBROUTINE SPLCN(X,Y,M,C)
C     THIS SUBROUTINE CALCULATES THE CONSTANTS TO BE USED IN A SPLINE
C     FIT OF THE DATA
      DIMENSION X(50),Y(50),D(50),P(50),E(50),C(4,50),A(50,3),B(50),Z(50
     1)
      MM=M-1
      DO 2 K=1,MM
      D(K)=X(K+1)-X(K)
      P(K)=D(K)/6.
    2 E(K)=(Y(K+1)-Y(K))/D(K)
      DO 3 K=2,MM
    3 B(K)=E(K)-E(K-1)
      A(1,2)=-1.-D(1)/D(2)
      A(1,3)=D(1)/D(2)
      A(2,3)=P(2)-P(1)*A(1,3)
      A(2,2)=2.*(P(1)+P(2))-P(1)*A(1,2)
      A(2,3)=A(2,3)/A(2,2)
      B(2)=B(2)/A(2,2)
      DO 4 K=3,MM
      A(K,2)=2.*(P(K-1)+P(K))-P(K-1)*A(K-1,3)
      B(K)=B(K)-P(K-1)*B(K-1)
      A(K,3)=P(K)/A(K,2)
    4 B(K)=B(K)/A(K,2)
      Q=D(M-2)/D(M-1)
      A(M,1)=1.+Q+A(M-2,3)
      A(M,2)=-Q-A(M,1)*A(M-1,3)
      B(M)=B(M-2)-A(M,1)*B(M-1)
      Z(M)=B(M)/A(M,2)
      MN=M-2
      DO 6 I=1,MN
      K=M-I
    6 Z(K)=B(K)-A(K,3)*Z(K+1)
      Z(1)=-A(1,2)*Z(2)-A(1,3)*Z(3)
      DO 7 K=1,MM
      Q=1./(6.*D(K))
      C(1,K)=Z(K)*Q
      C(2,K)=Z(K+1)*Q
      C(3,K)=Y(K)/D(K)-Z(K)*P(K)
    7 C(4,K)=Y(K+1)/D(K)-Z(K+1)*P(K)
      RETURN
      END
```

APPENDIX – Concluded

```
      SUBROUTINE MGAUSS(A,B,N,SUM,FUNC,FOFX,NUMBER)

      DIMENSION U(5),R(5),SUM(1),FOFX(1)
      DO 1 LL=1,NUMBER
    1 SUM(LL)=0.0
      IF(A.EQ.B) RETURN
      U(1)=.4256628305091 84
      U(2)=.2833023029353 76
      U(3)=.1602952158504 88
      U(4)=.0674683166555 08
      U(5)=.0130467357414 14
      R(1)=.1477621123573 76
      R(2)=.1346333596549 98
      R(3)=.1095431812579 91
      R(4)=.0747256745752 90
      R(5)=.0333567215434 4
      FINE=N
      DELTA=FINE/(B-A)
      DO 3 K=1,N
      XI =K-1
      FINE=A+XI/DELTA
      DO 2 II=1,5
      UU=U(II)/DELTA+FINE
      CALL FUNC (UU,FOFX)
      DO 2 JOYBOY=1,NUMBER
    2 SUM(JOYBOY)=R(II)*FOFX(JOYBOY)+SUM(JOYBOY)
      DO 3 JJ=1,5
      UU=(1.0-U(JJ))/DELTA+FINE
      CALL FUNC (UU,FOFX)
      DO 3 NN=1,NUMBER
    3 SUM(NN)=R(JJ)*FOFX(NN)+SUM(NN)
      DO 7 IJK=1,NUMBER
    7 SUM(IJK)=SUM(IJK)/DELTA
      RETURN
      END
```

REFERENCES

1. Langley, R. W.; and Billings, M. P.: A New Model for Estimating Space Proton Dose to Body Organs. Nucl. Technol., vol. 15, no. 1, July 1972, pp. 68-74.

2. Billings, M. P.; and Langley, R. W.: A Technique for Evaluation of Space Radiation Dose to Distributed Body Organs. Trans. American Nucl. Soc., vol. 14, no. 1, June 1971, pp. 416-417.

3. Billings, M. P.; and Yucker, W. R.: The Computerized Anatomical Man (CAM) Model. NASA CR-134043, 1973.

4. Wilson, J. W.; and Khandelwal, G. S.: Proton Tissue Dose Estimation in Complex Geometries. Trans. American Nucl. Soc., vol. 17, 1973, pp. 577-578.

5. Wilson, John W.; and Khandelwal, G. S.: Proton Dose Approximation in Arbitrary Convex Geometry. Nucl. Technol., vol. 23, no. 3, Sept. 1974, pp. 298-305.

6. Abramowitz, Milton; and Stegun, Irene A., eds.: Handbook of Mathematical Functions. Dover Publ., Inc., 1965.

TABLE 1.- AREAL DENSITY DISTRIBUTION FUNCTIONS

THICKNESS	ARMS	LEGS	HI TRUNK	LO TRUNK	SKULL	AVE BFO
1.100E+00	7.103E-03	0.	3.593E-03	0.	0.	1.910E-03
1.325E+00	5.301E-02	0.	9.344E-03	0.	0.	9.708E-03
1.595E+00	4.081E-02	0.	2.023E-02	0.	1.679E-03	1.101E-02
1.920E+00	9.993E-02	2.710E-02	2.509E-02	0.	4.205E-02	3.259E-02
2.300E+00	6.035E-02	2.555E-02	3.456E-02	0.	9.156E-02	3.315E-02
2.750E+00	6.911E-02	2.787E-02	2.289E-02	0.	9.277E-02	3.218E-02
3.300E+00	4.181E-02	4.881E-02	4.525E-02	1.630E-03	7.080E-02	3.967E-02
4.000E+00	3.819E-02	6.489E-02	4.426E-02	3.661E-03	4.395E-02	4.241E-02
4.800E+00	5.504E-02	6.281E-02	5.158E-02	1.465E-02	5.738E-02	4.914E-02
5.750E+00	1.537E-01	8.093E-02	4.724E-02	3.669E-02	5.282E-02	7.137E-02
6.950E+00	9.516E-02	7.954E-02	5.384E-02	4.808E-02	6.498E-02	6.777E-02
8.350E+00	5.825E-02	6.136E-02	4.914E-02	4.926E-02	8.854E-02	5.765E-02
1.005E+01	3.009E-02	5.231E-02	6.091E-02	4.352E-02	6.374E-02	5.074E-02
1.200E+01	1.144E-02	4.462E-02	3.726E-02	5.013E-02	4.053E-02	3.887E-02
1.450E+01	1.222E-02	2.843E-02	3.678E-02	5.675E-02	3.939E-02	3.470E-02
1.750E+01	1.092E-02	2.519E-02	2.413E-02	3.657E-02	2.767E-02	2.534E-02
2.100E+01	1.689E-02	1.424E-02	2.039E-02	3.117E-02	7.206E-03	1.888E-02
2.550E+01	1.077E-02	7.770E-03	1.531E-02	1.445E-02	2.145E-03	1.097E-02
3.050E+01	8.712E-03	5.058E-03	8.604E-03	8.333E-03	1.752E-03	6.840E-03
3.650E+01	4.127E-03	4.533E-03	4.387E-03	6.745E-03	1.679E-03	4.630E-03
4.400E+01	3.239E-03	1.842E-03	2.446E-03	3.661E-03	1.703E-03	2.528E-03
5.300E+01	1.266E-03	1.135E-03	1.377E-03	1.953E-03	9.804E-04	1.360E-03
6.350E+01	9.025E-04	7.524E-04	1.116E-03	6.823E-04	5.270E-04	8.348E-04
7.600E+01	5.167E-04	9.648E-04	4.908E-04	1.875E-04	6.985E-04	6.081E-04
9.150E+01	0.	0.	0.	0.	0.	0.

TABLE 2.- DOSES FOR SAMPLE SPECTRUM

	ARMS	LEGS	HI TRUNK	LO TRUNK	SKULL	AVE BFO
REM	7.5E-14	2.4E-14	3.2E-14	4.3E-15	4.1E-14	3.0E-14
RAD	4.9E-14	1.7E-14	2.0E-14	3.3E-15	2.8E-14	2.0E-14